
Overview for this session
Basic rules for ERC20 contracts

▶ Presentation: writing and debugging rules
▶ transfer changes balances appropriately
▶ transfer reverts when it should
▶ transfer doesn’t revert unexpectedly

▶ Exercise: similar rules for transferFrom

Generalized (parametric) rules

▶ Presentation: rules that apply to all methods
▶ Only the owner can increase their allowance
▶ The owner only changes their allowance deliberately

▶ Exercise: similar rules for balanceOf



ERC20 transfer and balanceOf
The first properties we’d like to test are described in the interface:
//// contracts/IERC20.spec

/**
* Interface of the ERC20 standard as defined in the EIP.
*/

interface IERC20 {

/**
* Moves `amount` tokens from the caller's account to `recipient`.
*/
function transfer(address recipient, uint256 amount)

external
returns(bool);

/**
* Returns the amount of tokens owned by `account`.
*/
function balanceOf(address account)

external view
returns(uint256);

...
}



Specifying transfer in CVL (unit-test-style rules)

//// certora/specs/ERC20.spec

methods {
balanceOf(address) returns (uint) envfree

}

/// Transfer must move `amount` tokens from
/// the caller's account to `recipient`.
rule transferSpec {

address sender; address recip; uint amount;

env e;
require e.msg.sender == sender;

mathint balance_sender_before = balanceOf(sender);
mathint balance_recip_before = balanceOf(recip);

transfer(e, recip, amount);

mathint balance_sender_after = balanceOf(sender);
mathint balance_recip_after = balanceOf(recip);

require sender != recip;

assert balance_sender_after == balance_sender_before - amount,
"transfer must decrease sender's balance by amount";

assert balance_recip_after == balance_recip_before + amount,
"transfer must increase recipient's balance by amount";

}

(results link)

(second link)

https://prover.certora.com/output/6554/7c3d8d9cd1e4a8d43246?anonymousKey=cd65d8676f1af617d565ba0e44d77a1f8ffd54a8 
https://prover.certora.com/output/6554/7a84047100c50672ef08?anonymousKey=eec3f306ae40b6c9d3b6288d041cbca9858ec394 


What about revert?
So far:

▶ Transfer reduces sender’s balance by amount
▶ Transfer increases recipient’s balance by amount

What if sender’s balance is less than amount?

▶ Transaction reverts
▶ No balances change!
▶ Why doesn’t this violate the rule?

Answer: by default, Prover ignores reverting paths.

▶ we can override this behavior to reason about reverting



transfer revert conditions

//// certora/specs/ERC20.spec

/// Transfer must revert if the sender's balance is too small
rule transferReverts {

env e; address recip; uint amount;

require balanceOf(e.msg.sender) < amount;

transfer@withrevert(e, recip, amount);

assert lastReverted,
"transfer(recip,amount) must revert if sender's balance is less than `amount`";

}

(results link)

Reasoning about reverts:

▶ Use f@withrevert(...) to consider paths where f reverts
▶ Use lastReverted to determine whether last call reverted

▶ Warning: it is always the last call!
▶ save it if you need tomake another call

https://prover.certora.com/output/6554/6f60ed5fb862a5d0875a?anonymousKey=00c547b1e751563fe338d494cc82705042e4ab3f 


Proving transfer doesn't revert (liveness rules)

//// certora/specs/ERC20.spec

/// Transfer must not revert unless
/// - the sender doesn't have enough funds
///
///
///
///
///
/// @title Transfer doesn't revert
rule transferDoesntRevert {

env e; address recipient; uint amount;

require balanceOf(e.msg.sender) > amount;

transfer@withrevert(e, recipient, amount);
assert !lastReverted;

}



Proving transfer doesn't revert (liveness rules)

//// certora/specs/ERC20.spec

/// Transfer must not revert unless
/// - the sender doesn't have enough funds
/// - or the message value is nonzero,
///
///
///
///
/// @title Transfer doesn't revert
rule transferDoesntRevert {

env e; address recipient; uint amount;

require balanceOf(e.msg.sender) > amount;
require e.msg.value == 0;

transfer@withrevert(e, recipient, amount);
assert !lastReverted;

}



Proving transfer doesn't revert (liveness rules)

//// certora/specs/ERC20.spec

/// Transfer must not revert unless
/// - the sender doesn't have enough funds
/// - or the message value is nonzero,
/// - or the recipient's balance would overflow,
///
///
///
/// @title Transfer doesn't revert
rule transferDoesntRevert {

env e; address recipient; uint amount;

require balanceOf(e.msg.sender) > amount;
require e.msg.value == 0;
require balanceOf(recipient) + amount < max_uint;

transfer@withrevert(e, recipient, amount);
assert !lastReverted;

}



Proving transfer doesn't revert (liveness rules)

//// certora/specs/ERC20.spec

/// Transfer must not revert unless
/// - the sender doesn't have enough funds
/// - or the message value is nonzero,
/// - or the recipient's balance would overflow,
/// - or the message sender is 0
///
///
/// @title Transfer doesn't revert
rule transferDoesntRevert {

env e; address recipient; uint amount;

require balanceOf(e.msg.sender) > amount;
require e.msg.value == 0;
require balanceOf(recipient) + amount < max_uint;
require e.msg.sender != 0;

transfer@withrevert(e, recipient, amount);
assert !lastReverted;

}



Proving transfer doesn't revert (liveness rules)

//// certora/specs/ERC20.spec

/// Transfer must not revert unless
/// - the sender doesn't have enough funds
/// - or the message value is nonzero,
/// - or the recipient's balance would overflow,
/// - or the message sender is 0
/// - or the recipient is 0
///
/// @title Transfer doesn't revert
rule transferDoesntRevert {

env e; address recipient; uint amount;

require balanceOf(e.msg.sender) > amount;
require e.msg.value == 0;
require balanceOf(recipient) + amount < max_uint;
require e.msg.sender != 0;
require recipient != 0;

transfer@withrevert(e, recipient, amount);
assert !lastReverted;

}

(results link)

https://prover.certora.com/output/6554/9a16f3d1434013167dc1?anonymousKey=6d74b9277ebc1760c3c5f03b52c1a82da7e50652 


Summary
▶ Writing rules is like writing unit tests

▶ But you can let the prover choose the values!

▶ Use mathint variables to avoid overflow in spec

▶ Pass env as first argument to specify msg.sender and other variables
▶ Use envfree declaration in methods block to avoid passing env

▶ By default, reverting paths are ignored
▶ Use @withrevert and lastReverted to reason about reverting paths
▶ Writing “liveness properties” is hard (but possible!)



Exercise (∼15 minutes)

So far (certora/specs/ERC20.spec):
▶ transferSpec
▶ transferReverts
▶ transferDoesntRevert

Exercise:
▶ Write transferFromSpec

▶ …get it to pass
▶ Try transferFromReverts
▶ Try transferFromSucceeds

To run:
sh certora/scripts/verifyERC20.sh

Ask for help!

//// contracts/IERC20.sol

/// Interface of the ERC20 standard as defined in the EIP.
interface IERC20 {

/// Moves `amount` tokens from `sender` to `recipient` using
/// the allowance mechanism. `amount` is then deducted from
/// the caller's allowance.
///
function transferFrom(

address sender,
address recipient,
uint256 amount

) external returns (bool);

/// Returns the remaining number of tokens that `spender`
/// will be allowed to spend on behalf of `owner` through
/// {transferFrom}.
///
/// This value changes when {approve} or {transferFrom} are
/// called.
///
function allowance(address owner, address spender)

external
view
returns(uint256);

}


